INTERNAL REPORTS IN

SIMULATION, OPTIMIZATION AND CONTROL

No. SOC-191

SUBROUTINES FOR IMPLEMENTING QUADRATIC MODELS OF SURFACES IN OPTIMAL DESIGN

H.L. Abdel-Malek and J.W. Bandler

April 1978

FACULTY OF ENGINEERING
McMASTER UNIVERSITY
HAMILTON, ONTARIO, CANADA

AVAILABILITY OF REPORTS

General Contributions of manuscripts to the collection of SOC reports are on the initiative of the authors. They are not solicited.

Catalog A catalog of SOC reports including abstracts, descriptions of contents and indication of related work is available on request with orders of $10.00 or more (payable in advance).

Charges To offset preparation, printing and distribution costs the nominal charges as indicated must be made. There is a minimum charge of $10.00 per mail order. A minimum subscription order of $100.00 will entitle the subscriber to past or future reports up to the accumulated value of his subscription. A minimum order of $500.00 will entitle the subscriber to a 50% discount, however, excessive shipment expenses are to be borne by the subscriber.

Ordering Any number of these reports may be ordered. (Special reduced rates will be quoted for multiple copies of the same report). Cheques should be made out in U.S. or Canadian dollars and made payable to "McMaster University". Requests must be addressed to:

Dr. J.W. Bandler
Coordinator, G-SOC
Faculty of Engineering
McMaster University
Hamilton, Canada L8S 4L7

Reports will not normally be sent out until payment is received. Any official requisition or order not accompanied by the correct advance payment will be returned to sender.

Restrictions Reports may at any time be restricted for internal use only. Some may be revised, withdrawn or superceded. Availability is not guaranteed and descriptions or charges are subject to change without notice. The right is reserved to substitute a similar report or reprint containing essentially the same technical material in lieu of any particular request unless written instructions to the contrary are explicity supplied.

Responsibility Neither authors of reports nor any member of the Group will accept any responsibility for the consequences of any use to which the reports themselves or their contents, in particular the computer programs, are put once they leave McMaster University. No responsibility to supply any future revisions or corrections of any kind will be entertained. No responsibility will be accepted for material lost in transit, not arriving by a specified date, or arriving in a damaged or unusable condition.

SUBROUTINES FOR IMPLEMENTING QUADRATIC MODELS OF SURFACES IN OPTIMAL DESIGN

H.L. Abdel-Malek and J.W. Bandler

Abstract

This collection of subroutines is primarily for approximating functions of many variables by quadratic polynomials as well as evaluating them. They are oriented for use in a tolerance optimization process. The quadratic approximation is obtained, exploiting sparsity, in Subroutine MODEL4. The evaluation of the resulting polynomial at the vertices of the tolerance orthotope is performed using an efficient technique.

This work was supported by the National Research Council of Canada under Grant A7239 and by a Postdoctorate Fellowship to H.L. Abdel-Malek.

The authors are with the Group on Simulation, Optimization and Control and Department of Electrical Engineering, McMaster University, Hamilton, Canada L8S 4L7.

DESCRIPTION

This report should be read in conjunction with previous work by the authors [1-3]. The quadratic interpolation is performed by solving a system of linear equations with the coefficients of the quadratic polynomial as unknowns. In general, the resulting system of linear equations is of order

$$N = (k+1)(k+2)/2,$$

where k is the number of parameters [1].

For a particular choice of base points (points where the quadratic polynomial coincides with the approximated function), the effort in solving the system of simultaneous linear equations can be reduced to $5k^2 - 2k$ operations (multiplications or divisions) instead of the well known $(N^3 + 3N^2 - N)/3$ operations for Gauss elimination. This is done by exploiting sparsity in the system of linear equations in Subroutine MODEL4.

The evaluation of the quadratic polynomial and its derivatives using Subroutine QPE is performed in an efficient manner. The procedure is to start by evaluating the polynomial and its derivatives at the vertex nearest to the origin. Values at other vertices are obtained from neighbouring ones exploiting simple properties of quadratic polynomials [1,2].

Subroutine VERT generates the vectors $\underset{\sim}{\mu}^r$ for each vertex $\underset{\sim}{\phi}^r$ [1,2], according to the vertex numbering scheme

$$r = 1 + \sum_{i=1}^{k} \left(\frac{\mu_i^r + 1}{2} \right) 2^{i-1} .$$

REFERENCES

[1] J.W. Bandler and H.L. Abdel-Malek, "Optimal centering, tolerancing and yield determination using multidimensional approximations", Proc. 1977 IEEE Int. Symp. Circuits and Systems (Phoenix, AZ, April 1977), pp. 219-222.

[2] J.W. Bandler and H.L. Abdel-Malek, "Optimal centering, tolerancing and yield determination via updated approximations and cuts", Faculty of Engineering, McMaster University, Hamilton, Canada, Report SOC-173, June 1977 (Revised Dec. 1977).

[3] H.L. Abdel-Malek and J.W. Bandler, "Yield optimization for arbitrary statistical distributions", Proc. 1978 IEEE Int. Symp. Circuits and Systems (New York, NY, May 1978).

APPENDIX

FORTRAN LISTING OF THE SUBROUTINES MODEL4, QPE AND VERT

```
      SUBROUTINE MODEL4(N,NR,NOBP,X,B,V,CX,AF,TAF,H,MP,NREM,IJ)
C     =========================================================
C
C         N = NUMBER OF PARAMETERS
C
C         NR = NUMBER OF FUNCTIONS TO BE INTERPOLATED
C
C         NOBP = NUMBER OF BASE POINTS = (N+1)*(N+2)/2
C
C         X = ARRAY OF DIMENSION N IN WHICH CENTER OF INTERPOLATION IS
C             GIVEN
C
C         B = AN ARRAY OF DIMENSION NOBP * NR IN WHICH THE COEFFICIENTS
C             OF THE POLYNOMIALS WILL BE RETURNED, FOR EXAMPLE, B(5,2) IS
C             THE 5TH COEFFICIENT OF THE 2ND POLYNOMIAL
C
C         NREM = N*(N-1)
C
C         V = AN ARRAY OF DIMENSIONS NREM IN WHICH THE USER SUPPLIES
C             NUMBERS SUCH THAT ABS(V) LESS THAN ONE AND NOT EQUAL TO
C             ZERO. THESE NUMBERS ARE USED TO CHANGE TWO PARAMETERS AT
C             A TIME
C
C         CX = WORRKING ARRAY OF DIMENSION N TO STORE THE CENTER
C              OF INTERPOLATION
C
C         AF = AN ARRAY OF DIMENSION NR USED AS A WORKING AREA TO STORE
C              ACTUAL FUNCTION VALUES AT A CERTAIN BASE POINT
C
C         TAF = WORKING ARRAY OF DIMENSION NR USED FOR STORING 2*AF
C
C         H = AN ARRAY OF DIMENSION N IN WHICH THE STEPS DEFINING THE
C             SIZE OF THE INTERPOLATION REGION ARE SUPPLIED BY THE USER
C
C         MP = 1 IF THE QUADRATIC MODEL IS TO BE PRINTED, OTHERWISE SET
C              TO ZERO
C
C         IJ = 1 FOR FIRST CALL TO MODEL4, OTHERWISE SET TO ZERO
C
C         SUBROUTINE ACFUNS(X,AF) IS CALLED TO SUPPLY VALUES OF AF(NR)
C         AT X(N)
C
C - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - -
C
C
C         *******    E X A M P L E    *******
C
C             THIS EXAMPLE ILLUSTRATES HOW TO OBTAIN THE COEFFICIENTS OF
C             THE INTERPOLATING POLYNOMIAL USING MODEL4
C
C     PROGRAM TST( INPUT,OUTPUT,PUNCH,TAPE5=INPUT,TAPE6=OUTPUT
C    +,TAPE7=PUNCH)
C     DIMENSION X(3),CX(3),B(10,1),V(6),AF(1),TAF(1),H(3)
C     DATA N,NR,NOBP,MP,NREM,IJ/3,1,10,1,6,1/
C     DATA H/3*0.04/
C     DATA V/-0.9,-0.7,0.6,0.5,0.5,-0.6/
C     DATA X/2.,2.,1./
C
C     CALL MODEL4(N,NR,NOBP,X,B,V,CX,AF,TAF,H,MP,NREM,IJ)
C     PUNCH(7,100) (B(I,1),I=1,NOBP)
C 100 FORMAT(5E16.8)
C     STOP
C     END
C
C
C     SUBROUTINE ACFUNS(X,AF)
C
C     DIMENSION X(1),AF(1)   . . . . .
C
C     AF(I) = FUNCTION OF X(1,2,..,N)   , I=1,2,..NR
C
C     RETURN
C     END
C
```

```
C        -------------------------------------------------------------
C
      DIMENSION X(N),B(NOBP,NR),V(NREM),CX(N),AF(NR),TAF(NR)
      DIMENSION H(N)
C
      IF(IJ.EQ.0) GO TO 10
      MM=((N+2)*(N+1))/2
      MM1=MM-1
      MMN=MM-N-1
      LK=N-1
   10 CALL SECOND(T1)
      DO 20 J=1,N
      CX(J)=X(J)
   20 CONTINUE
      CALL ACFUNS(X,AF)
      DO 25 J=1,NR
      B(MM,J)=AF(J)
      TAF(J)=AF(J)+AF(J)
   25 CONTINUE
      DO 40 I=1,N
      X(I)=CX(I)+H(I)
      CALL ACFUNS(X,AF)
      DO 30 J=1,NR
      B(I,J)=AF(J)
   30 CONTINUE
      X(I)=CX(I)-H(I)
      CALL ACFUNS(X,AF)
      TH=H(I)+H(I)
      HH=H(I)*H(I)
      THH=HH+HH
      DO 35 J=1,NR
      B(MMN+I,J)=(B(I,J)-AF(J))/TH
      B(I,J)=(B(I,J)+AF(J)-TAF(J))/THH
   35 CONTINUE
      X(I)=CX(I)
   40 CONTINUE
      IR=N
      IRR=1
      DO 50 I=1,LK
      I1=I+1
      MI=MMN+I
      DO 50 K=I1,N
      VHI=V(IRR)*H(I)
      VHK=V(IRR+1)*H(K)
      MK=MMN+K
      IR=IR+1
      IRR=IRR+2
      X(I)=CX(I)+VHI
      X(K)=CX(K)+VHK
      DV=VHI*VHK
      CALL ACFUNS(X,AF)
      DO 45 J=1,NR
      B(IR,J)=(AF(J)-B(MM,J)-VHI*(VHI*B(I,J)+B(MI,J))
     +                -VHK*(VHK*B(K,J)+B(MK,J)))/DV
   45 CONTINUE
      X(I)=CX(I)
      X(K)=CX(K)
   50 CONTINUE
      DO 65 J=1,NR
      DO 55 I=1,N
      MI=MMN+I
      A=B(I,J)*CX(I)
      B(MM,J)=B(MM,J)+CX(I)*(A-B(MI,J))
      B(MI,J)=B(MI,J)-A-A
   55 CONTINUE
      IR=N
      DO 60 I=1,LK
      MI=MMN+I
      I1=I+1
      DO 60 K=I1,N
      MK=MMN+K
      IR=IR+1
      A=CX(I)*B(IR,J)
      B(MM,J)=B(MM,J)+A*CX(K)
```

```
      B(MI,J)=B(MI,J)-CX(K)*B(IR,J)
      B(MK,J)=B(MK,J)-A
   60 CONTINUE
   65 CONTINUE
      CALL SECOND(T2)
C
      IF(MP.NE.1) RETURN
      WRITE(6,200)
      DO 70 I=1,N
      WRITE(6,250) H(I)
   70 CONTINUE
      T=T2-T1
      WRITE(6,260) T
      DO 150 MA=1,MR
      WRITE(6,270) MA
      WRITE(6,280) (B(I,MA),I=1,MM)
  150 CONTINUE
  200 FORMAT(1H1,4X,*STEPS AROUND CENTER OF INTERPOLATION*)
  250 FORMAT(4X,E16.8)
  260 FORMAT(/,4X,*MODELING TIME IN SECONDS =*,E16.8)
  270 FORMAT(/,4X,*ER(*,I2,*)*,/,4X,6(*-*))
  280 FORMAT(5(4X,E16.8),/)
  311 FORMAT(5E16.8)
      RETURN
      END
```

```
      SUBROUTINE QPE(N,NN,NAC,X,ER,GE,B,IV,H,Y,NOBP,NV,LL)
C     ====================================================
C
C        N = NUMBER OF VARIABLES ( INCLUDING TOLERANCES)
C
C        NN = NUMBER OF NOMINAL VALUES OF THE PARAMETERS  (OR NUMBER
C             OF VARIABLES IN CASE OF A SINGLE POINT)
C
C        NAC = NUMBER OF ACTUAL CONSTRAINTS  , E.G.,IF THERE ARE FOUR
C               VERTICES FOR TWO CONSTRAINTS NAC=2, 8 ERROR FUNCTIONS
C             RESULT
C
C        X = INPUT VECTOR OF DIMENSION N
C
C        ER = ARRAY OF DIMENSION NAC IF LL=0, OTHERWISE (LL=1) OF
C             DIMENSION NV*NAC
C
C        GE = ARRAY OF GRADIENTS OF ERROR FUNCTIONS HAS DIMENSION (N,M),
C             WHERE M IS THE NUMBER OF ERROR FUNCTIONS AS INDICATED IN
C             THE DIMENSION OF ER
C
C        B = ARRAY OF POLYNOMIAL COEFFICIENTS, DIMENSIONED (NOBP,NAC),
C            WHERE  NOBP = (NN+1)*(NN+2)/2
C
C        IV = ARRAY DEFINING THE VERTICES AND IS SUPPLIED BY THE USER,
C             DIMENSIONED (NN,NV), IF EVALUATION AT A SINGLE POINT IS
C             REQUIRED (LL=0) THEN NV=1 AND IV IS NOT SUPPLIED
C
C        H = WORKING ARRAY OF DIMENSION (NN,NN)
C
C        Y = WORKING ARRAY OF DIMENSION NN
C
C        NOBP = (NN+1)*(NN+2)/2     SUPPLIED BY USER
C
C                NN
C        NV = 2     NUMBER OF VERTICES AND EQUALS 1 FOR A SINGLE POINT
C
C        LL = INTEGER SUPPLIED BY USER, FOR LL=1 EVALUATION AT VERTICES
C             IS REQUIRED, FOR LL=0 EVALUATION AT A SINGLE POINT IS
C             REQUIRED
C
C  - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - - -
C
C          *******    E X A M P L E    *******
C
C          THIS EXAMPLE ILLUSTRATES THE USE OF QPE TO SUPPLY THE VALUES
C          OF FIVE POLYNOMIALS AT THE VERTICES FOR A PROBLEM WITH THREE
C          PARAMETERS
C              X(I),  I=1,2,3       ARE THE NOMINAL PARAMETER VALUES
C              X(I),  I=4,5,6       ARE THE CORRESPONDING TOLERANCES
C
C          THE VALUES OF B(10,5) AND IV(3,8) ARE ASSUMED TO BE
C          TRANSFERED THROUGH A COMMON STATEMENT
C
C     DIMENSION ER(40),GE(6,40),H(3,3),Y(3)
C     COMMON/MIM/ B(10,5),IV(3,8)
C     DATA N,NN,NAC,NOBP,NV,LL/6,3,5,10,8,1/
C     DATA X/1.1,2.0,3.0,0.01,0.5,0.15/
C
C     CALL QPE(N,NN,NAC,X,ER,GE,B,IV,H,Y,NOBP,NV,LL)
C
C     WRITE(6,*) ER,GE
C     STOP
C     END
C
C
C      -------------------------------------------------------------------
C
C
      DIMENSION X(1),ER(1),GE(N,1),B(NOBP,NAC),IV(NN,NV),H(NN,NN),Y(NN)
      MM=NOBP
      MM1=MM-1
      LK=NN-1
C
```

```
      IF(LL.EQ.1) GO TO 10
      DO 5 I=1,NN
      Y(I)=X(I)
    5 CONTINUE
      GO TO 25
C
C         STARTING AT THE VERTEX Y(I)=X(I)-X(I+NN)       X(I+NN)  IS EPS(I)
C
   10 DO 20 I=1,NN
      Y(I)=X(I)-X(I+NN)
   20 CONTINUE
   25 DO 100 K1=1,NAC
      K=1+NV*(K1-1)
      IF(LL.NE.1) K=K1
      S=B(MM,K1)
      DO 30 I=1,NN
      T=B(I,K1)*Y(I)
      S=S+T*Y(I)
      GE(I,K)=T+T
      H(I,I)=B(I,K1)+B(I,K1)
   30 CONTINUE
      KM=NN
      DO 40 I=1,LK
      LH=I+1
      DO 40 J=LH,NN
      KM=KM+1
      T=B(KM,K1)*Y(I)
      S=S+T*Y(J)
      GE(I,K)=GE(I,K)+B(KM,K1)*Y(J)
      GE(J,K)=GE(J,K)+T
      H(I,J)=B(KM,K1)
      H(J,I)=H(I,J)
   40 CONTINUE
      DO 50 I=1,NN
      KM=KM+1
      S=S+B(KM,K1)*Y(I)
      GE(I,K)=GE(I,K)+B(KM,K1)
   50 CONTINUE
      ER(K)=S
      IF(LL.NE.1) GO TO 100
C
C              OTHER VERTICES
C
      JU=1
      L=K
      LG=1
      MI=K-1
      DO 90 I=1,NN
      M=MI
      INN=I+NN
      EP=X(INN)
      TE=EP+EP
      DO 70 IR=1,NN
      H(IR,I)=TE*H(IR,I)
   70 CONTINUE
      EPH=EP*H(I,I)
      DO 80 J=1,JU
      M=M+1
      L=L+1
      LG=LG+1
      ER(L)=ER(M)+TE*GE(I,M)+EPH
      DO 80 IR=1,NN
      GE(IR,L)=GE(IR,M)+H(IR,I)
   80 CONTINUE
      JU=JU+JU
   90 CONTINUE
      L=K
      DO 99 II=1,NV
      DO 95 I=1,NN
      J=I+NN
      GE(J,L)=FLOAT(IV(I,II))*GE(I,L)
   95 CONTINUE
      L=L+1
   99 CONTINUE
```

```
100 CONTINUE                                              QPE1480
    RETURN                                                QPE1490
    END                                                   QPE1500
```

```
      SUBROUTINE VERT(IV,NN,NV)
C     ========================
C
C         THIS SUBROUTINE GENERATES THE  MU S  ACCORDING TO THE COMMON
C         NUMBERING SCHEME OF THE VERTICES
C
C       NN = NUMBER OF PARAMETERS (INPUT)
C       NV = NUMBER OF VERTICES (INPUT)
C       IV = ARRAY OF  MUS  (OUTPUT)
C
C    ------------------------------------------------------------------
C
      DIMENSION IV(NN,NV)
C
      II=1
      MI=NV
      DO 30 I=1,NN
      IR=0
      IS=-1
      DO 20 J=1,MI
      DO 10 K=1,II
      IR=IR+1
      IV(I,IR)=IS
   10 CONTINUE
      IS=-IS
   20 CONTINUE
      II=II+II
      MI=MI/2
   30 CONTINUE
      RETURN
      END
```

SOC-191

SUBROUTINES FOR IMPLEMENTING QUADRATIC MODELS OF SURFACES IN OPTIMAL DESIGN

H.L. Abdel-Malek and J.W. Bandler

April 1978, No. of Pages: 11

Revised:

Key Words: Surface fitting, quadratic approximation, polynomial evaluation

Abstract: This collection of subroutines is primarily for approximating functions of many variables by quadratic polynomials as well as evaluating them. They are oriented for use in a tolerance optimization process. The quadratic approximation is obtained, exploiting sparsity, in Subroutine MODEL4. The evaluation of the resulting polynomial at the vertices of the tolerance orthotope is performed using an efficient technique.

Description: Contains Fortran Listing, user's manual. The listing contains 355 cards, of which 164 are comment cards.

Related Work: SOC-118, SOC-173, SOC-182, SOC-184, SOC-185.

Price: $15.00.